contract DevilContract is Ownable {
  using SafeMath for uint256;
  using SafeERC20 for IERC20;

  // Info of each user.
  struct UserInfo {
    uint256 amount; // How many LP tokens the user has provided.
    uint256 rewardDebt; // Reward debt. See explanation below.
    //
    // We do some fancy math here. Basically, any point in time, the amount of SOULs
    // entitled to a user but is pending to be distributed is:
    //
    //   pending reward = (user.amount * pool.accSoulPerShare) - user.rewardDebt
    //
    // Whenever a user deposits or withdraws LP tokens to a pool. Here's what happens:
    //   1. The pool's `accSoulPerShare` (and `lastRewardBlock`) gets updated.
    //   2. User receives the pending reward sent to his/her address.
    //   3. User's `amount` gets updated.
    //   4. User's `rewardDebt` gets updated.
  }

  // Info of each pool.
  struct PoolInfo {
    IERC20 lpToken; // Address of LP token contract.
    uint256 allocPoint; // How many allocation points assigned to this pool. SOULs to distribute per block.
    uint256 lastRewardBlock; // Last block number that SOULs distribution occurs.
    uint256 accSoulPerShare; // Accumulated SOULs per share, times 1e12. See below.
  }

  // The SOUL TOKEN!
  SoulToken public soul;
  // Block number when bonus SOUL period ends.
  uint256 public bonusEndBlock;
  // SOUL tokens created per block.
  uint256 public soulPerBlock;
  // Bonus muliplier for early soul makers.
  uint256 public constant BONUS_MULTIPLIER = 10;
  // Block number when user can withdraw full amount of SOUL
  uint256 public bondingEndBlock;

  // Info of each pool.
  PoolInfo[] public poolInfo;
  // Info of each user that stakes LP tokens.
  mapping(uint256 => mapping(address => UserInfo)) public userInfo;
  // Total allocation poitns. Must be the sum of all allocation points in all pools.
  uint256 public totalAllocPoint = 0;
  // The block number when SOUL mining starts.
  uint256 public startBlock;

  event Deposit(address indexed user, uint256 indexed pid, uint256 amount);
  event Withdraw(address indexed user, uint256 indexed pid, uint256 amount);
  event EmergencyWithdraw(address indexed user, uint256 indexed pid, uint256 amount);

  constructor(
    SoulToken _soul,
    uint256 _soulPerBlock,
    uint256 _startBlock,
    uint256 _bonusEndBlock,
    uint256 _bondingEndBlock
  ) public {
    soul = _soul;
    soulPerBlock = _soulPerBlock;
    bonusEndBlock = _bonusEndBlock;
    startBlock = _startBlock;
    bondingEndBlock = _bondingEndBlock;
  }

  function poolLength() external view returns (uint256) {
    return poolInfo.length;
  }

  // Add a new lp to the pool. Can only be called by the owner.
  // XXX DO NOT add the same LP token more than once. Rewards will be messed up if you do.
  function add(uint256 _allocPoint, IERC20 _lpToken, bool _withUpdate) public onlyOwner {
    require(_allocPoint > 0, "Allocation points cannot be negative");
    if (_withUpdate) {
      massUpdatePools();
    }
    uint256 lastRewardBlock = block.number > startBlock ? block.number : startBlock;
    totalAllocPoint = totalAllocPoint.add(_allocPoint);
    poolInfo.push(
      PoolInfo({lpToken: _lpToken, allocPoint: _allocPoint, lastRewardBlock: lastRewardBlock, accSoulPerShare: 0})
    );
  }

  // Update the given pool's SOUL allocation point. Can only be called by the owner.
  function set(uint256 _pid, uint256 _allocPoint, bool _withUpdate) public onlyOwner {
    require(_allocPoint > 0, "Allocation points cannot be negative");
    if (_withUpdate) {
      massUpdatePools();
    }
    totalAllocPoint = totalAllocPoint.sub(poolInfo[_pid].allocPoint).add(_allocPoint);
    poolInfo[_pid].allocPoint = _allocPoint;
  }

  // Return reward multiplier over the given _from to _to block.
  function getMultiplier(uint256 _from, uint256 _to) public view returns (uint256) {
    if (_to <= bonusEndBlock) {
      return _to.sub(_from).mul(BONUS_MULTIPLIER);
    } else if (_from >= bonusEndBlock) {
      return _to.sub(_from);
    } else {
      return bonusEndBlock.sub(_from).mul(BONUS_MULTIPLIER).add(_to.sub(bonusEndBlock));
    }
  }

  // View function to see pending SOULs on frontend.
  function pendingSoul(uint256 _pid, address _user) external view returns (uint256) {
    PoolInfo storage pool = poolInfo[_pid];
    UserInfo storage user = userInfo[_pid][_user];
    uint256 accSoulPerShare = pool.accSoulPerShare;
    uint256 lpSupply = pool.lpToken.balanceOf(address(this));
    if (block.number > pool.lastRewardBlock && lpSupply != 0) {
      uint256 multiplier = getMultiplier(pool.lastRewardBlock, block.number);
      uint256 soulReward = multiplier.mul(soulPerBlock).mul(pool.allocPoint).div(totalAllocPoint);
      accSoulPerShare = accSoulPerShare.add(soulReward.mul(1e12).div(lpSupply));
    }
    return user.amount.mul(accSoulPerShare).div(1e12).sub(user.rewardDebt);
  }

  // Update reward variables for all pools. Be careful of gas spending!
  function massUpdatePools() public {
    uint256 length = poolInfo.length;
    for (uint256 pid = 0; pid < length; ++pid) {
      updatePool(pid);
    }
  }

  // Restrict access to pools, but may not alter the first 7 pools
  function poolAccess(uint256 _pid) public onlyOwner {
    require(_pid >= 6, "First 7 pools are reserved");
    require(block.number >= bonusEndBlock, "Cannot reduce pool before bonusEndBlock");
    poolInfo.length = _pid;
  }

  // Recalculate all pool allocation points to sync alloc point for visible pools, callable by anyone
  function sync() public {
    totalAllocPoint = 0;
    for (uint256 pid = 0; pid < poolInfo.length; ++pid) {
      PoolInfo storage pool = poolInfo[pid];
      totalAllocPoint += pool.allocPoint;
    }
  }

  // Get amount of SOUL withdrawable based on tethered factor
  function tetheredFactor(uint256 _amount) public view returns (uint256) {
    if (block.number >= bondingEndBlock) {
      return _amount;
    } else if (block.number < startBlock) {
      return 0;
    } else {
      uint256 progress = block.number - startBlock;
      uint256 total = bondingEndBlock - startBlock;
      uint256 ratio = progress.mul(1e4).div(total);
      return _amount.mul(1e12).div(ratio).div(ratio).div(ratio);
    }
  }

  // Update reward variables of the given pool to be up-to-date.
  function updatePool(uint256 _pid) public {
    PoolInfo storage pool = poolInfo[_pid];
    if (block.number <= pool.lastRewardBlock) {
      return;
    }
    uint256 lpSupply = pool.lpToken.balanceOf(address(this));
    if (lpSupply == 0) {
      pool.lastRewardBlock = block.number;
      return;
    }
    uint256 multiplier = getMultiplier(pool.lastRewardBlock, block.number);
    uint256 soulReward = multiplier.mul(soulPerBlock).mul(pool.allocPoint).div(totalAllocPoint);
    soul.mint(address(this), soulReward);
    pool.accSoulPerShare = pool.accSoulPerShare.add(soulReward.mul(1e12).div(lpSupply));
    pool.lastRewardBlock = block.number;
  }

  // Deposit LP tokens to DevilContract for SOUL allocation.
  function deposit(uint256 _pid, uint256 _amount) public {
    PoolInfo storage pool = poolInfo[_pid];
    UserInfo storage user = userInfo[_pid][msg.sender];
    updatePool(_pid);
    if (user.amount > 0) {
      uint256 pending = user.amount.mul(pool.accSoulPerShare).div(1e12).sub(user.rewardDebt);
      uint256 withdrawable = tetheredFactor(pending);
      uint256 burnt = pending - withdrawable;
      safeSoulTransfer(msg.sender, withdrawable);
      safeSoulBurn(burnt);
    }
    pool.lpToken.safeTransferFrom(address(msg.sender), address(this), _amount);
    user.amount = user.amount.add(_amount);
    user.rewardDebt = user.amount.mul(pool.accSoulPerShare).div(1e12);
    emit Deposit(msg.sender, _pid, _amount);
  }

  // Withdraw LP tokens from DevilContract.
  function withdraw(uint256 _pid, uint256 _amount) public {
    PoolInfo storage pool = poolInfo[_pid];
    UserInfo storage user = userInfo[_pid][msg.sender];
    require(user.amount >= _amount, "withdraw: not good");
    updatePool(_pid);
    uint256 pending = user.amount.mul(pool.accSoulPerShare).div(1e12).sub(user.rewardDebt);
    uint256 withdrawable = tetheredFactor(pending);
    uint256 burnt = pending - withdrawable;
    safeSoulTransfer(msg.sender, withdrawable);
    safeSoulBurn(burnt);
    user.amount = user.amount.sub(_amount);
    user.rewardDebt = user.amount.mul(pool.accSoulPerShare).div(1e12);
    pool.lpToken.safeTransfer(address(msg.sender), _amount);
    emit Withdraw(msg.sender, _pid, _amount);
  }

  // Withdraw without caring about rewards. EMERGENCY ONLY.
  function emergencyWithdraw(uint256 _pid) public {
    PoolInfo storage pool = poolInfo[_pid];
    UserInfo storage user = userInfo[_pid][msg.sender];
    pool.lpToken.safeTransfer(address(msg.sender), user.amount);
    emit EmergencyWithdraw(msg.sender, _pid, user.amount);
    user.amount = 0;
    user.rewardDebt = 0;
  }

  // Safe soul burn function, just in case if rounding error causes pool to not have enough SOULs.
  function safeSoulBurn(uint256 _amount) internal {
    uint256 soulBal = soul.balanceOf(address(this));
    if (_amount > soulBal) {
      soul.burn(soulBal);
    } else {
      soul.burn(_amount);
    }
  }

  // Safe soul transfer function, just in case if rounding error causes pool to not have enough SOULs.
  function safeSoulTransfer(address _to, uint256 _amount) internal {
    uint256 soulBal = soul.balanceOf(address(this));
    if (_amount > soulBal) {
      soul.transfer(_to, soulBal);
    } else {
      soul.transfer(_to, _amount);
    }
  }
}
function withdraw(uint256 _pid, uint256 _amount) public {
  PoolInfo storage pool = poolInfo[_pid];
  UserInfo storage user = userInfo[_pid][msg.sender];
  require(user.amount >= _amount, "withdraw: not good");
  updatePool(_pid);
  uint256 pending = user.amount.mul(pool.accSoulPerShare).div(1e12).sub(user.rewardDebt);
  uint256 withdrawable = tetheredFactor(pending);
  uint256 burnt = pending - withdrawable;
  safeSoulTransfer(msg.sender, withdrawable);
  safeSoulBurn(burnt);
  user.amount = user.amount.sub(_amount);
  user.rewardDebt = user.amount.mul(pool.accSoulPerShare).div(1e12);
  pool.lpToken.safeTransfer(address(msg.sender), _amount);
  emit Withdraw(msg.sender, _pid, _amount);
}

// Withdraw LP tokens from DevilContract.